T-603-THYD: Compiler Project - Parser

Due on October 19, 2016 at 23:59

Professor David Gudni Halldorsson TA

Darri Steinn Konradsson

Problem 1

Change the grammar by eliminating left recursion and left factor the grammar.

Solution

program ::= **class** **id** **{** *variable_declarations* *method_declarations* **}**

variable_declarations ::= *type* *variable_list* ; *variable_declarations* | ϵ

type ::= **int** | **real**

variable_list ::= *variable* *variable_list'*

variable_list' ::= , *variable* *variable_list'* | ϵ

variable ::= **id** *variable'*

variable' ::= [**num**] | ϵ

method_declarations ::= *method_declaration* *more_method_declarations*

more_method_declarations ::= *method_declaration* *more_method_declarations* | ϵ

method_declaration ::= **static** *method_return_type* **id** (*parameters*)
{ *variable_declarations* *statement_list* **}**

method_return_type ::= *type* | **void**

parameters ::= *parameter_list* | ϵ

parameter_list ::= *type* **id** *parameter_list'*

parameter_list' ::= , *type* **id** *parameter_list'* | ϵ

statement_list ::= *statement* *statement_list* | ϵ

statement ::= *statement'* ;
| **if** (*expression*) *statement_block* *optional_else*
| **for** (*variable_loc* = *expression* ; *expression* ; *incr_decr_var*
) *statement_block*
| *statement_block*

statement' ::= *variable_loc* = *expression*
| **id** (*expression_list*)
| **return** *optional_expression*
| **break**
| **continue**
| *incr_decr_var*

optional_expression ::= *expression* | ϵ

statement_block ::= { *statement_list* }

incr_decr_var ::= *variable_loc* **incdecop**

optional_else ::= **else** *statement_block* | ϵ

expression_list ::= *expression* *more_expressions* | ϵ

more_expressions ::= , *expression* *more_expressions* | ϵ

expression ::= *simple_expression* *expression'*

expression' ::= **relop** *simple_expression* | ϵ

simple_expression ::= *term* *simple_expression'* | *sign* *term* *simple_expression'*

simple_expression' ::= **addop** *term* *simple_expression'* | ϵ

term ::= *factor* *term'*

term' ::= **mulop** *factor* *term'* | ϵ

factor ::= *variable_loc*
| **id** (*expression_list*)
| **num**
| (*expression*)
| ! *factor*

variable_loc ::= **id** *variable_loc'*

variable_loc' ::= [*expression*] | ϵ

sign ::= + | −